Assignment 1 - *Discrete Models* - *Profir Alexandru* - *ICA* 246 — 2

1) Find the solution for the following difference equations:

$(a)\ x_{n+1} = \left(\dfrac{n+1}{n+2}\right)^2 \cdot x_n + \dfrac{1}{n+2}, x_0 = 1;$

$(b)\ x_{n+3} - 4 \cdot x_{n+2} + x_{n+1} + 6 \cdot x_n = 60 \cdot 4^n, x_0 = 2, x_1 = 12, x_2 = 12;$

$(c)\ x_{n+1} = \dfrac{2 \cdot x_n}{1 + 4 \cdot x_n}, x_0 = 1 \left(\mathit{Hint} : \textbf{use}\ \mathit{substitution}\ x_n = \dfrac{1}{y_n} \right);$

```
> restart:

  eq := x(n+1) = ((n+1)/(n+2))^2 * x(n) + 1/(n+2);
  ans := factor(simplify(rsolve({eq, x(0)=1}, x(n))));
```

$$eq := x(n+1) = \frac{(n+1)^2 x(n)}{(n+2)^2} + \frac{1}{n+2}$$

$$ans := \frac{n+2}{2\ (n+1)} \tag{1}$$

```
> restart:

  eq := x(n+3) - 4 * x(n+2) + x(n+1) + 6 * x(n) = 60 * 4^n;
  ans := factor(simplify(rsolve({eq, x(0)=2, x(1)=12, x(2)=12}, x
  (n))));
```

$$eq := x(n+3) - 4\,x(n+2) + x(n+1) + 6\,x(n) = 60\ 4^n$$

$$ans := -16\ 3^n + 16\ 2^n - 4\ (-1)^n + 6\ 4^n \tag{2}$$

```
> restart:

  eq := x(n+1) = (2*x(n)) / (1 + 4 * x(n));

  # Subbing x(n) with 1/y(n)
  x0 := 1; y0 := 1/x0;
  eq_y := simplify(subs(x(n)=1/y(n), x(n+1)=1/y(n+1), eq));
  eq_y := y(n+1) = solve(eq_y, y(n+1));

  ans_y := rsolve({eq_y, y(0)=y0}, y(n));
  ans_x := factor(simplify(1/ans_y));
```

$$eq := x(n+1) = \frac{2\,x(n)}{1 + 4\,x(n)}$$

$$x0 := 1$$

$$y0 := 1$$

$$eq\_y := \frac{1}{y(n+1)} = \frac{2}{y(n) + 4}$$

$$eq\_y := y(n+1) = \frac{y(n)}{2} + 2$$

$$ans\_y := -3\left(\frac{1}{2}\right)^n + 4$$

$$ans\_x := -\frac{1}{3\,2^{-n} - 4} \quad \mathbf{(3)}$$

2) Let us consider the difference equation:

$$x_{n+1} = \frac{x_n^2 + 7}{2 \cdot x_n}$$

(a) Find the equilibrium points and study their stability.
(b) Do some numerical simulations.

(a) The equilibrium points can be found using the 'solve' utility of Maple. However, given the non-linearity of the difference equation,
the stability of the equilibrium points can be studied using the 'Stability in the first approximation' Theorem, based on 'Taylor Series Expansion'

$$f(x) = f(a^*) + f'(a^*) \cdot (x - a^*) + f''(a^*) \cdot \left(\frac{(x - a^*)^2}{2}\right) + \ldots$$

Given the existence of an equilibrium point $a^* = f(a^*)$, and $f'(a^*)$ exists, then:

$(i)\ If\,|f'(a^*)| < 1 \Rightarrow a^*\ is\ locally\ asimptotically\ stable$

$(ii)\ If\ |f'(a^*)| > 1 \Rightarrow a^*\ is\ unstable$

Given the theory support above, and $x_1^* = \sqrt{7}, x_2^* = -\sqrt{7}$, and $f'(x) = \frac{x^2 - 7}{2\,x^2}$, both

derivate values in the equilibrium points are < 1 in absolute value,
thus both equilibrium points are locally asimptotically stable.
(b) Numerical simulations are done below.

```
> restart:
  with(plots):

  f := x -> (x^2+7) / (2 * x);

  xstar1, xstar2 := solve(x=f(x), x);
  evalf(xstar1); evalf(xstar2);

  fdx := simplify(D(f)(x));
  abs_fdx_xstar1 := abs(subs(x=xstar1, fdx));
  abs_fdx_xstar2 := abs(subs(x=xstar2, fdx));

  n := 100;
  xxpos[1] := 10;
```

```
xxneg[1] := -10;

for i from 1 to n - 1 do
    xxpos[i+1] := evalf(f(xxpos[i]));
    xxneg[i+1] := evalf(f(xxneg[i]));
end do:

ppos := pointplot([seq([k, xxpos[k]], k=1..n)],
                symbol=solidcircle, color=blue):

pneg := pointplot([seq([k, xxneg[k]], k=1..n)],
                symbol=solidcircle, color=red):

display(ppos, pneg, view=[0..n, -10..10]);
```

$$f := x \mapsto \frac{x^2+7}{2\,x}$$

$$xstar1, xstar2 := \sqrt{7}, -\sqrt{7}$$

$$2.645751311$$

$$-2.645751311$$

$$fdx := \frac{x^2-7}{2\,x^2}$$

$$abs\_fdx\_xstar1 := 0$$

$$abs\_fdx\_xstar2 := 0$$

$$n := 100$$

$$xxpos_1 := 10$$

$$xxneg_1 := -10$$



3) Let us consider the system of difference equations:

$$\begin{cases} x_{n+1} = x_n - x_n^2 - x_n y_n \\ y_{n+1} = 2\,y_n - y_n^2 - 3\,x_n y_n \end{cases}$$

(a) Find the equilibrium points and study their stability.
(b) Do some numerical simulations.

(a) Given the non-linearity of the system, the equilibrium points can be found using the linearized system
resulted from the Jacobian of the system.

$$y_{n+1} = Jacobian(f)\left(x^*\right) y_n$$

Theorem: The following statements hold:

$(i)\, x^*$ *is locally asimptotically stable* **if and** *only* **if** $|\lambda| < 1$,

for *all* λ *eigenvalues of the Jacobian*(f) (x^*);

(ii) x^* *is unstable* **if** *there exists an eigenvalue* λ *of the Jacobian*(f) (x^*) *such that* $|\lambda| > 1$.

Equilibrium points

- $x=\frac{1}{2}, y=-\frac{1}{2}; |\lambda_{1,2}| > 1 \Rightarrow$ *unstable*;
- $x=0, y=0; |\lambda_2=2| > 1 \Rightarrow$ *unstable*;
- $x=0, y=1; |\lambda_{1,2}| < 1 \Rightarrow$ *locally asimptotically stable*;

(b) Numerical simulations are done below.

```
> restart:
  with(plots): with(linalg):

  f1 := (x, y) -> x - x^2 - x*y;
  f2 := (x, y) -> 2*y - y^2 - 3*x*y;

  ans_sys := solve({f1(x,y)=x, f2(x,y)=y}, {x, y});
  J := jacobian([f1(x,y), f2(x,y)], [x,y]);

  # 1st solution: {x=1/2, y=-1/2}
  A1 := subs(x=1/2, y=-1/2, evalm(J));
  lambda1, lambda2 := evalf(eigenvals(A1)):
  abs(lambda1); abs(lambda2);

  # 2nd solution: {x=0, y=0}
  A2 := subs(x=0, y=0, evalm(J));
  lambda1, lambda2 := evalf(eigenvals(A2)):
  abs(lambda1); abs(lambda2);

  # 3rd solution: {x=0, y=1}
  A3 := subs(x=0, y=1, evalm(J));
  lambda1, lambda2 := evalf(eigenvals(A3)):
  abs(lambda1); abs(lambda2);

  # simulation
  n := 20;
  xx[1] := 0.1:
  yy[1] := 0.1:

  for i from 1 to n-1 do
      xx[i+1] := evalf(f1(xx[i], yy[i]));
      yy[i+1] := evalf(f2(xx[i], yy[i]));
  end do:

  pts_x := [seq([k, xx[k]], k=1..n)]:
  pts_y := [seq([k, yy[k]], k=1..n)]:

  p1 := pointplot(pts_x, symbol=circle, color=red):
  p2 := pointplot(pts_y, symbol=cross,  color=blue):

  display(p1, p2, title="x (red), y (blue) vs iteration",
          view=[0..n, -1..2]);
```

$$f1 := (x, y) \mapsto x - x^2 - x\,y$$

$$f2 := (x, y) \mapsto 2\,y - y^2 - 3\,x\,y$$

$$ans\_sys := \left\{x = \frac{1}{2}, y = -\frac{1}{2}\right\}, \{x = 0, y = 0\}, \{x = 0, y = 1\}$$

$$J := \begin{bmatrix} -2\,x - y + 1 & -x \\ -3\,y & -3\,x - 2\,y + 2 \end{bmatrix}$$

$$A1 := \begin{bmatrix} \frac{1}{2} & -\frac{1}{2} \\ \frac{3}{2} & \frac{3}{2} \end{bmatrix}$$

$$1.224744871$$

$$1.224744871$$

$$A2 := \begin{bmatrix} 1 & 0 \\ 0 & 2 \end{bmatrix}$$

$$1.$$

$$2.$$

$$A3 := \begin{bmatrix} 0 & 0 \\ -3 & 0 \end{bmatrix}$$

$$0.$$

$$0.$$

$$n := 20$$



4) Let's consider the simple interest and compound interest models:

$$S_{n+1} = S_n + pS_0$$

$$S_{n+1} = S_n + \frac{p}{r} S_0$$

Company A offers simple interest at an annual rate of 4%. Company B offers compound interest at an annual rate of 3% with a conversion period of one month.

(a) Calculate for the both cases the amount on deposit after 5, 10, 15, and 20 years for principal $S_0 = 1000$.

The results are displayed in the following format: [years, amount_on_deposit, rate_conversion]

- Company A: [5, 1200.00, 1.20], [10, 1400.00, 1.40], [15, 1600.00, 1.60], [20, 1800.00, 1.80];
- Company B: [5, 1161.61, 1.16], [10, 1349.35, 1.34], [15, 1567.43, 1.56], [20, 1820.75, 1.82];

(b) Determine which interest offer maximizes the amount on deposit after 5, 10, 15, and 20 years.

- Company A maximizes the amount on deposit after 5, 10, and 15 years;
- Company B maximizes the amount on deposit after 20 years;

The offer of company A is profitable for acceptable subscription periods, while company B is profitable (and not by a great margin) for longer periods.

The plots provided below confirm the presented conclusion.

```
> restart:
  with(plots):

  S0 := 1000;
  years := [5, 10, 15, 20];

  SIM := S(n+1) = S(n) + p * S0;
  CIM := S(n+1) = S(n) * (1 + (p/r));

  SIM_ans:= unapply(factor(simplify(rsolve({SIM, S(0)=S0}, S(n)))),
  p, n);
  CIM_ans := unapply(factor(simplify(rsolve({CIM, S(0)=S0}, S(n)))
  ), p, n, r):
  CIM_ans_years := (p, n, r) -> CIM_ans(p, r*n, r);

  SIM_A := [seq([y, SIM_ans(0.04, y), SIM_ans(0.04, y) / S0], y in
  years)];
  CIM_B := [seq([y, CIM_ans_years(0.03, y, 12), CIM_ans_years(0.03,
  y, 12) / S0], y in years)];

  plot([
        [seq([SIM_A[i][1], SIM_A[i][2]], i=1..nops(years))],
        [seq([CIM_B[i][1], CIM_B[i][2]], i=1..nops(years))]
       ],
       style = pointline, color = [red, blue, green],
       legend = ["Simple 4%", "Compound 3% monthly"],
       title = "Comparison of interest growth",
       labels = ["Years", "Amount ($)"]);

  plot([
        [seq([SIM_A[i][1], SIM_A[i][3]], i=1..nops(years))],
        [seq([CIM_B[i][1], CIM_B[i][3]], i=1..nops(years))]
       ],
       style = pointline, color = [red, blue],
       legend = ["Simple 4%", "Compound 3% monthly"],
       title = "Comparison of rate conversion",
       labels = ["Years", "Amount (%)"]);
```

$$S0 := 1000$$

$$years := [5, 10, 15, 20]$$

$$SIM := S(n+1) = S(n) + 1000\,p$$

$$CIM := S(n+1) = S(n)\left(1 + \frac{p}{r}\right)$$

$$SIM\_ans := (p, n) \mapsto 1000\,n\,p + 1000$$

$$CIM\_ans\_years := (p, n, r) \mapsto CIM\_ans(p, r\,n, r)$$

$SIM\_A := [[5, 1200.00, 1.200000000], [10, 1400.00, 1.400000000], [15, 1600.00, 1.600000000], [20, 1800.00, 1.800000000]]$

$CIM\_B := [[5, 1161.616782, 1.161616782], [10, 1349.353547, 1.349353547], [15, 1567.431725, 1.567431725], [20, 1820.754995, 1.820754995]]$





Comparison of rate conversion
1.8
1.7
1.6
1.5
1.4
1.3
1.2
Amount (%)
5
10
15
20
Years
Simple 4%
Compound 3% monthly